

Dopo avervi presentato un programma gestionale (Archivio indirizzi - MC n. 19) e due programmi svagherecci (Domino e Tombola - MC n. 20) questo mese vogliamo offrirvi l'opportunità di diventare milionari o, perché no, miliardari facendo girare sul vostro TI 99 il programma Totocalcio di Mario Laganà.

Lo sappiamo che il campionato è appena finito, già presentando la tombola vi abbiamo detto che ci piace arrivare in anticipo ...

Prima di andarci ad impelagare nella "descrizione descrittiva" del listato, rubiamo un po' di spazio alla rubrica per una breve precisazione.


Nel paragrafo considerazioni generali dell'articolo sul programma Archivio indirizzi di Marina Travaglini (MC n. 19 pag. 77) ci sono due piccoli errori che per amor di precisione non potevamo far finta di ignorare anche se comunque non pregiudicano il buon funzionamento dell'archivio.

Il programma è costituito da 210 righe (non 211) e la somma delle lunghezze dei campi del singolo record non deve superare i 128 caratteri (non i 192).

Si incorre in simili sviste perché a volte i listati vengono modificati all'ultimo momento al fine di ottimizzarli ulteriormente e nella maggior parte dei casi quando sia il testo dell'articolo che le figure di illustrazione sono già pronte con i riferimenti relativi alla versione precedente.

Ad esempio nella foto del primo quadro video

del programma Totocalcio presentato questo mese viene indicato un numero massimo di colonne pari a 16, mentre in realtà esaminando ulteriormente il listato e facendolo girare sul Texas, abbiamo trovato un paio di banchi che ci hanno costretto a limitare tale numero a 14. Per fortuna questa volta siamo riusciti ad avvertirvi tempestivamente.



Totocalcio

Mario Laganà - Taranto

È possibile trovare in commercio parecchi programmi di compilazione schedine da inserire nei personal più diffusi.

Gli algoritmi di generazione dei pronostici per il totocalcio sono diversi; si passa senza soluzione di continuità dalla estrazione completamente casuale del simbolo 1,2,X alla complicatissima gestione di dati relativi ai risultati delle varie partite disputate negli ultimi dieci anni con riferimenti a parametri statistici che controllano il calcolo delle probabilità.

Si può tuttavia affermare che se gli ideatori di tali programmi si occupano della loro commercializzazione piuttosto che di come passare il tempo a Palm Beach, ciò vuol dire che la sicurezza matematica della vincita (e quindi della vacanza a Palm Beach) è più lontana della possibilità di ricavare qualche soldino vendendo tale software a delle persone che sperano a loro volta di abbandonare Cesenatico per un'isola della Polinesia.

Il programma che vi presentiamo usa un algoritmo di generazione casuale guidata, ossia per ogni partita dovete specificare quale percentuale di probabilità esiste che il risultato sia 1 e quale che sia X (il 2 viene

Totocalcio

```

10 REM ***** TOTOCALCIO DI MARIO LAGANA' *****
20 DIM H(182)
30 DIM VET(100)
40 RANDOMIZE
50 K=1
60 GOSUB 600
70 CALL CLEAR
80 CALL SCREEN(8)
90 PRINT "PRONOSTICI TOTOCALCIO"
100 PRINT "-----":
110 PRINT "QUANTE COLONNE (MAX 14)":
120 INPUT COLONNE
130 PRINT
140 COLONNE=INT(COLONNE)
150 IF COLONNE(<=14 THEN 210
160 PRINT "NUMERO TROPPO ALTO"
170 FOR I=1 TO 700
180 NEXT I
190 CALL CLEAR
200 GOTO 110
210 FOR PARTITA=1 TO 13
220 PRINT "PARTITA";PARTITA;TAB(12);"1, X IN % ":
230 INPUT UND, ICS
240 UND=INT(UND)
250 ICS=INT(ICS)
260 IF UND+ICS(<=100 THEN 290
270 CALL SOUND(2500,-1,4)
280 GOTO 220
290 FOR LO=1 TO COLONNE
300 PRD=INT(100*RND+1)
310 FOR L=1 TO 100
320 IF VET(L)=PRD THEN 340
330 NEXT L
340 IF L>UND+ICS THEN 380
350 IF L<=UND THEN 400
360 H(K)=0
370 GOTO 410
380 H(K)=2
390 GOTO 410
400 H(K)=1
410 K=K+1
420 NEXT LO
430 NEXT PARTITA
440 CALL CLEAR
450 PRINT "ECCO LA SCHEDINA !! ":
460 PRINT "-----":
470 REM ***** STAMPA LA SCHEDINA *****
480 FOR PRO=1 TO COLONNE*13
490 IF H(PRO)=0 THEN 530
500 IF H(PRO)=1 THEN 550
510 PRINT "2 ";
520 GOTO 560
530 PRINT "X ";
540 GOTO 560
550 PRINT "1 ";
560 IF INT(PRO/COLONNE) {} PRO/COLONNE THEN 580
570 PRINT
580 NEXT PRO
590 END
600 REM ***** MESCOLA I NUMERI *****
610 FOR I=1 TO 100
620 VET(I)=I
630 NEXT I
640 FOR I=100 TO 2 STEP -1
650 J=INT(RND*I)+1
660 N=VET(I)
670 VET(I)=VET(J)
680 VET(J)=N
690 NEXT I
700 RETURN

```

VEETTORE PROBABILITA'

VEETTORE VET(100) PRIMA DELLA ROUTINE MESCOLO

1	2	3	4	5		98	99	100
1	2	3	4	5		98	99	100

VEETTORE VET(100) DOPO LA ROUTINE

1	2	3	4	5		98	99	100
32	9	76	54	98		11	3	69

VEETTORE PRONOSTICI

VEETTORE H(182) - 13 PARTITE X 14 COLONNE

1	2	3	4	5		180	181	182
2	0	1	1	0		2	2	1

PRONOSTICI TOTOCALCIO

```

QUANTE COLONNE <MAX 16> ? 8
P P P P P P P P P P P P P P P P
R R R R R R R R R R R R R R R R
T T T T T T T T T T T T T T T T
A A A A A A A A A A A A A A A A
1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
X X X X X X X X X X X X X X X X
I I I I I I I I I I I I I I I I
Z Z Z Z Z Z Z Z Z Z Z Z Z Z Z Z
N N N N N N N N N N N N N N N N
V V V V V V V V V V V V V V V V
N N N N N N N N N N N N N N N N
O O O O O O O O O O O O O O O O
V V V V V V V V V V V V V V V V
A A A A A A A A A A A A A A A A
O O O O O O O O O O O O O O O O

```

Inserimento percentuali probabilità per i risultati di ogni singola partita.

ECCO LA SCHEGINA !! :

```

Z I X X X X X X X X X X X X X X
N N N N N N N N N N N N N N N N
N N N N N N N N N N N N N N N N
N N N N N N N N N N N N N N N N
X X X X X X X X X X X X X X X X
N N N N N N N N N N N N N N N N
I I I I I I I I I I I I I I I I
X X X X X X X X X X X X X X X X
I I I I I I I I I I I I I I I I
** DONE **
>

```

Visualizzazione della schedina, in questo caso su otto colonne.

ricavato per differenza). In tal modo se indicate il valore 50 per l'1 ed il valore 40 per l'X avremo un 10% di possibilità che esca come risultato il 2. È ovvio che indicando il 100% per uno dei tre dati richiesti e lo 0% per i due rimanenti, il computer non potrà che generare il simbolo al quale abbiamo assegnato il totale delle probabilità.

Prima di passare alla descrizione del programma vogliamo farvi una raccomandazione: se per caso vi capitasse di fare tredici e di vincere un'ingente somma non siate tanto egoisti da dimenticarvi di chi lo ha ideato (Laganà) e di chi lo ha diffuso (MC!).

Analisi del listato

Il programma è costituito da 70 linee. Dopo l'usuale REM di intestazione troviamo il dimensionamento dei vettori H e VET di 182 e 100 elementi rispettivamente. Facciamo notare a tal proposito che dal momento che non è stata impiegata l'istruzione OPTION BASE 1 in realtà tali vettori sono costituiti da 183 e 101 elementi; quello identificato dal subscripto 0 non verrà utilizzato.

La linea 40 (RANDOMIZE) provvede a cambiare la sequenza di generazione dei numeri casuali della funzione RND che verrà impiegata nelle righe 300 e 650. Alla linea 600 inizia la routine di mescolamento

dei numeri da 1 a 100 che prima vengono inseriti progressivamente nel vettore VET (610-630) e quindi permutati casualmente tra di loro per mezzo di un loop di 99 cicli (640-690).

Prima di proseguire oltre nella descrizione delle singole istruzioni sarà opportuno spendere due parole sul principio logico di funzionamento in base al quale sia stato possibile ottenere una generazione casuale guidata da una percentuale di probabilità prestabilita.

Il fulcro di tale algoritmo è il vettore VET il quale dopo la procedura di mescolamento dei numeri da 1 a 100 che esso contiene ci permetterà di ottenere il risultato voluto. Nella linea 300 viene assegnato alla variabile PRO un valore casuale da 1 a 100, che viene subito dopo cercato sequenzialmente in VET; il subscripto relativo trovato viene confrontato con i valori delle percentuali di probabilità assegnati ai simboli 1, X, 2 e quindi, in base al risultato ottenuto, verrà memorizzato nel vettore H il numero 1, 0 o 2 dove 0 sostituisce momentaneamente il valore alfanumerico X. Se per esempio abbiamo assegnato al segno 1 il 50% delle probabilità, all'X il 40% e al 2 il 10% si otterrà l'1 se il subscripto trovato sarà minore o uguale a 50, l'X se sarà compreso tra 51 e 90 o, infine, il 2 se maggiore di 90. Il vettore H è stato dimensionato a 182 elementi in quanto dovrà

contenere provvisoriamente tutti i segni generati che saranno poi visualizzati sullo schermo al termine del programma; dal momento che le partite sono 13 ed il numero massimo delle colonne previste è pari a 14 ecco che $13 \times 14 = 182$.

Le linee 440-580 si incaricano di mostrare la schedina ottenuta per un numero di colonne pari a quelle impostate inizialmente; da notare il modo elegante in cui alla riga 560 si risolve il problema di come stabilire quando si siano visualizzati sulla stessa riga tutti i risultati relativi alla singola partita evitando l'uso di un loop nidificato.

Utilizzazione del programma

Non c'è molto da dire a tale riguardo dal momento che tutto quello che occorre indicare è il numero delle colonne desiderate (massimo 14) ed i valori percentuali, separati da una virgola, relativi ai segni 1 e X per ogni partita. Il programma prevede delle istruzioni di controllo che non accettano valori illeciti che lo possano mandare in "tilt"; sarebbe infatti assurdo assegnare il 60% delle probabilità al segno 1 ed il 50% al segno X in quanto la loro somma supererebbe il 100%.

Il tempo di esecuzione è direttamente proporzionale al numero delle colonne impiegate a causa, è ovvio, del maggior numero di calcoli richiesti. **MC**